
회 사 에 서  바 로  통 하 는  실 무  엑 셀  매 크 로 & V B A

VBA로 엑셀의 모든 기능을 코드로 구현할 수 있지만, 실제 업무에 사용하는 코드는 한정되어 

있습니다. 따라서 실무에서 많이 사용되는 VBA 학습에서 하나의 완성된 프로그램을 제작하는 

것은 앞으로 VBA를 활용해 업무에 적용하는 데 있어 중요한 경험이 됩니다. 특별 부록으로 제

공되는 고급 실무 프로젝트를 통해 하나의 프로그램을 처음부터 끝까지 직접 따라 하며 제작해 

VBA 프로그래밍 감각을 경험할 수 있습니다.

실무 프로그램
제작으로

VBA 프로그래밍
실전 감각
기르기

사용자 정의 폼, 레이블, 텍스트 상자, 명령 단추, Workbook_Open 이벤트, Workbook_BeforeClose 
이벤트, For Next 반복문, For Each 반복문, If 조건문, Cells 속성, Visible 속성, Save 메서드, Show 
메서드, Unload 문

로그인 정보를 확인한 후 파일을
열어주는 보안 프로그램 만들기

실습 파일 | 별책부록/인사기록부보안파일.xlsx   

완성 파일 | 별책부록/인사기록부보안파일(완성).xlsm

회사에서
바로 통하는
키워드

01
고급 실무 프로젝트

01   프로젝트 시작하기

엑셀에서 통합 문서 열기 암호를 설정하면 문서를 보호할 수 있지만 암호는 한 개만 사용할 수 있습니다. 

여러 사람이 공동으로 사용하는 통합 문서라면 로그인 아이디와 비밀번호를 각각 다르게 사용하여 문서 

열기 권한을 부여할 수 있도록 하면 편리합니다. 

인사기록부 보안 파일을 열면 매크로가 실행되면서 아이디와 비밀번호를 확인하는 사용자 정의 폼이 나타

나고, 이 폼에서 입력하는 아이디와 비밀번호를 통합 문서에 입력해 둔 정보와 비교한 후 일치할 경우 [인

사기록부] 시트를 표시하는 프로그램을 만들어보겠습니다. 만약 입력하는 아이디와 비밀번호가 일치하지 

않으면 보안 파일인 [인사기록부] 시트를 표시하지 않고 안내 메시지가 나타나도록 하겠습니다.

02   프로젝트 예제 미리 보기

한눈에 
보는 
작업순서

준비된 시트
확인하기

통합 문서가 열릴 때
폼 보여주는 프로시저 작성하기

사용자 정의 폼
추가하기

폼에 컨트롤
추가하기

로그인 정보를 확인하는
프로시저 작성하기

통합 문서가 닫히기 전에
시트 숨기는 프로시저 작성하기

03   핵심기능 미리 보기

STEP 02 로그인 정보를 확인하는 이벤트 프로시저 작성하기

1  로그인 폼에서 [로그인] 단추를 클릭하면 실행되는 Click 이벤트 프로시저를 추가합니다.

2  필요한 변수를 선언하고 변수에 초깃값을 대입합니다.

3   로그인 폼에서 입력한 아이디와 비밀번호가 [로

그인정보] 시트의 데이터와 일치하는지 확인하

는 조건문과 반복문을 입력합니다.

4   정보가 일치하면 [인사기록부] 시트와 [로그인정

보] 시트를 숨기기 취소하고, [시작화면] 시트는 

숨기는 코드를 입력합니다.

5   정보가 일치하지 않으면 메시지 대화상자를 표

시하고 폼을 닫는 코드를 입력합니다.

STEP 01 시트 준비와 사용자 정의 폼 작성하기

1   통합 문서가 열렸을 때 기본으로 보여줄 시트와 로그인 정보를 확인한 후에 보여줄 보안 문서 

시트를 준비합니다.

2   로그인 정보로 아이디와 비밀번호를 입력할 수 있는 사용자 정의 폼을 삽입하고, 레이블과 텍스

트 상자, 명령 단추를 폼에 추가합니다.

STEP 03 통합 문서가 열리고 닫히기 직전에 프로그램이 실행되도록 이벤트 프로시저 작성하기

1   통합 문서가 열릴 때 폼이 실행되도록 ‘Work 

book_Open’ 이벤트 프로시저를 작성합니다.

2   통합 문서가 닫히기 직전에 [인사기록부] 시트와 

[로그인정보] 시트를 숨기고, [시작화면] 시트만 

표시하도록 ‘Workbook_BeforeClose’ 이벤트 

프로시저를 작성합니다. 

시트 준비와 사용자 정의 폼 작성하기01
STEP

이 통합 문서에는 매크로가 실행되지 않은 상태에서 보여줄 시트와 로그인 정보를 확인한 후 보여줄 시

트가 구분되어 있습니다. 로그인 정보를 확인하는 프로그램을 만들었다 하더라도 매크로 보안 경고에서 

[콘텐츠 사용]을 클릭하지 않으면 프로그램이 의미가 없기 때문에 매크로가 제외된 상태에서 문서를 열

었을 때는 중요한 시트를 볼 수 없도록 숨겨두기 위해서입니다. 준비된 시트를 확인하고 로그인 정보를 

입력하는 사용자 정의 폼을 작성합니다.

실습 파일을 열고 1  [시작화면] 시트 탭을 클릭합니다. 이 시트는 매크로가 실

행되지 않았을 때 기본으로 보여줄 시트로, 매크로가 실행되고 로그인 정보가 확인되면 숨겨집니

다. 2  [인사기록부] 시트 탭을 클릭합니다. 이 시트는 로그인 정보가 확인된 후 보여줄 시트로, 파

일이 닫힐 때는 VBA 코드에 의해 다시 숨겨집니다. 

01 시트 확인하기 

실력향상 시트를 숨길 때 Visible 속성값을 xlSheetVeryHidden으로 설정하면 시트 탭을 마우스 오른쪽 버튼으로 클릭하여 

[숨기기 취소]를 클릭하는 작업으로는 숨기기 취소를 할 수 없습니다. VBA 코드로 Visible 속성을 다시 xlSheetVisible로 변경해야 시

트를 표시할 수 있습니다.

1

2

[로그인정보] 시트 탭을 클릭합니다. 이 시트에는 로그인이 가능한 아이디와 비밀번호 목록이 

입력되어 있고 새로운 사용자가 있거나 로그인 정보를 변경할 때는 이 시트에서 직접 수정할 수 있

습니다. 이 시트도 로그인 정보가 확인된 후 보여줄 시트이며 파일이 닫힐 때는 VBA 코드에 의해 

다시 숨겨집니다. 

02  

1  [Visual Basic 편집기]에서 [삽입]-[사용자 정의 폼]을 클릭합니

다. 2  [속성] 창에서 [이름]에 frm로그인을, 3  [Caption]에 사용자 확인을 입력합니다.  
03 사용자 정의 폼 삽입하기 

실력향상 속성 창이 표시되지 않으면 [보기] - [속성 창]을 클릭합니다.

1

2 입력

3 입력

1  폼 오른쪽 하단 모서리의 크기 조절점을 드래그하여 폼 크기를 변경

합니다. 2  [도구 상자]에서 [레이블 ], [텍스트 상자 ], [명령 단추 ]를 각각 클릭하여 폼에 추

가한 후 각 컨트롤의 속성을 다음과 같이 변경합니다. 

04 폼에 컨트롤 추가하기 

1
2

3

4

5

번호 컨트롤 종류 속성 변경

1 레이블 •Caption : 아이디, 비밀번호

2 텍스트 상자 •이름 : txtID

3 텍스트 상자
•이름 : txtPW
•PasswordChar : *

4 명령 단추
•이름 : cmd로그인
•Caption : 로그인

5 명령 단추
•이름 : cmd취소
•Caption : 취 소

1  [표준] 도구 모음에서 [Sub/사용자 정의 폼 실행 ]을 클릭합니다. 폼이 실행

됩니다. 2  [닫기 ]를 클릭하여 폼을 닫습니다. 
05 폼 실행하기 

1 드래그

2

1

2

로그인 정보를 확인하는 이벤트 프로시저 작성하기02
STEP

로그인 폼에서 아이디와 비밀번호를 입력한 후 [로그인]을 클릭하면 [로그인정보] 시트에 입력된 데이터

와 일치하는지 확인하는 이벤트 프로시저를 작성해보겠습니다. 로그인 정보가 일치할 경우 [인사기록

부] 시트와 [로그인정보] 시트를 화면에 표시하고, [시작화면] 시트는 숨기기 취소합니다. 만약 로그인 정

보가 일치하지 않으면 메시지 대화상자를 표시하고 폼을 닫습니다.

[frm로그인] 폼에 추가한 [로그인] 명령 단

추를 더블클릭합니다. [코드] 창에 ‘cmd로그인_Click’ 이벤트 프로시저가 추가됩니다.
06 로그인 단추의 Click 이벤트 프로시저 작성하기 

폼에 입력하는 아이디와 비밀번호가 [로그인정보] 시트의 내용과 일치하는지 확인하고 일치

하면 [인사기록부] 시트와 [로그인정보] 시트를 보여주는 조건문과 반복문을 다음과 같이 입력합

니다.

07  

입력

1    Private Sub cmd로그인_Click()

2    Dim 행번호 As Integer

3    Dim 로그인시트 As Worksheet

4    Dim 시트 As Worksheet

5    Set 로그인시트 = Sheets("로그인정보")

6    For 행번호 = 2 To 로그인시트.Range("A1").End(xlDown).Row

7        If 로그인시트.Cells(행번호, 1).Value = Me.txtID.Value And _

                  로그인시트.Cells(행번호, 2).Value = Val(Me.txtPW.Value) Then

8             MsgBox "아이디와 비밀번호가 확인되었습니다"        

9             For Each 시트 In Worksheets

10                  If 시트.Name <> "시작화면" Then 시트.Visible = xlSheetVisible

11             Next        

12             Sheets("시작화면").Visible = xlSheetHidden

13             Sheets("인사기록부").Select

14             Unload Me

15             Exit Sub

16        End If

17    Next

18    MsgBox "로그인 정보가 일치하지 않습니다"

19    Unload Me

End Sub

1   [로그인] 명령 단추를 클릭했을 때 실행되는 이벤트 프로시저입니다.

2   로그인 정보를 확인할 때 사용할 반복문 카운터 변수를 선언합니다.

3   [로그인정보] 시트를 저장할 ‘로그인시트’ 변수를 선언합니다.

4   현재 통합 문서의 모든 시트를 저장하여 반복할 때 사용할 ‘시트’ 변수를 선언합니다.

5   ‘로그인시트’ 변수에 [로그인정보] 시트를 대입합니다.

6   반복문을 시작하여 [로그인정보] 시트의 A열에 입력된 셀 데이터 개수만큼 7  ~ 16을 반복 실행합니다. 

반복문의 시작값은 2이고, 최종값은 로그인시트.Range(“A1”).End(xlDown).Row에 의해 [로그인] 

시트의 [A1] 셀에서 아래쪽으로 데이터가 입력된 범위의 행 개수를 계산하여 결정합니다.

7   정보가 일치하는지 확인하는 조건문을 시작합니다. 한 문장인데 줄 언더바(_)를 사용하여 두 줄로 입력

했습니다. [로그인시트]의 A열 데이터와 폼의 아이디 입력란의 값이 일치하고 B열 데이터와 폼의 [비밀

번호] 입력란의 값이 일치하는지 확인합니다. 텍스트 상자에 입력되는 값은 문자로 인식되므로 Val 함

수를 사용하여 숫자 형식으로 변환한 후 비교합니다. 조건식이 참이면 8  ~ 15를 실행하고 거짓이면 17

에서 Next를 만나 6으로 이동하여 반복횟수가 남아 있으면 다음 정보를 다시 비교합니다.

8   로그인 정보가 확인되었다는 메시지 대화상자를 나타냅니다.

9   ~ 11  [시작화면] 시트를 제외한 나머지 시트를 모두 표시하는 반복문입니다. 현재 통합 문서의 모든 워

크시트를 ‘시트’ 변수에 대입한 후 시트 이름이 ‘시작화면’이 아니면 숨기기 취소합니다. If 조건문은 한 

줄로 입력하여 End If가 없습니다.

12   [시작화면] 시트를 숨기기합니다. Visible의 속성값으로 xlSheetHidden을 사용하면 시트 탭을 마우

스 오른쪽 버튼으로 클릭하여 [숨기기 취소]를 할 수 있습니다.

13   [인사기록부] 시트를 선택합니다.

14   ‘frm로그인’ 폼을 닫습니다.

15   프로그램을 종료합니다. 로그인 정보가 확인되면 18  ~ 19를 실행하지 않아야 하므로 필요한 문장입니다.

16   If 문을 종료합니다. 

17   반복문을 종료합니다.

18   로그인 정보가 일치하지 않는다는 메시지 대화상자를 나타냅니다. 로그인 정보가 일치하지 않은 상태

에서 반복문이 종료된 후에 실행되는 문장입니다.

19   ‘frm로그인’ 폼을 닫습니다.

1  [개체]에서 [cmd취소]를 클릭합니다. ‘cmd

취소_Click’ 이벤트 프로시저가 추가됩니다. [취소]를 클릭했을 때 현재 폼이 닫히면 되므로 2  

Unload Me를 입력합니다. 

08 취소 단추의 Click 이벤트 프로시저 작성하기 

2 입력

1

1  [표준] 도구 모음에서 [Sub/사용자 정의 폼 실행 ]을 클릭합니다. [사용자 

확인] 대화상자의 2  [아이디]에 onwings, [비밀번호]에 1234를 입력하고 3  [로그인]을 클릭합

니다. 로그인 정보가 확인되었다는 메시지가 나타나면 4  [확인]을 클릭합니다. [시작화면] 시트는 

숨기기되고, [인사기록부] 시트와 [로그인정보] 시트만 표시됩니다.

09 결과 확인하기 

2 입력

1

3

4

[Visual Basic 편집기]의 [표준] 도구 모음에서 1  [Sub/사용자 정의 폼 실행 ]을 클릭합니다. 

[사용자 확인] 대화상자에 2  아이디와 비밀번호를 임의로 입력한 후 3  [로그인]을 클릭합니다. 로

그인 정보가 일치하지 않는다는 메시지가 나타납니다. 4  [확인]을 클릭합니다.

10  

실력향상 로그인 정보가 일치하지 않으면 [시작화면] 시트만 표시되고 다른 시트들은 모두 숨기기되어야 하는데 이 부분은 

‘Workbook_BeforClose’ 이벤트 프로시저에서 해결합니다. 

1

3

4

2 임의로 입력

통합 문서가 열리고 닫히기 직전에 프로그램이 실행되도록
이벤트 프로시저 작성하기

03
STEP

통합 문서가 열릴 때 폼이 실행되도록 ‘Workbook_Open’ 이벤트 프로시저를 작성하고, 통합 문

서가 닫히기 직전에 [인사기록부] 시트와 [로그인정보] 시트를 숨기고 [시작화면] 시트만 표시하도록 

‘Workbook_BeforClose’ 이벤트 프로시저를 작성해보겠습니다.  

[Visual Basic 편집기]의 [프로젝트] 탐색

기 창에서 1  [현재_통합_문서]를 더블클릭합니다. [현재_통합_문서 (코드)] 창이 표시되면 2  [개체]

에서 [Work book]을 클릭합니다. ‘Workbook_Open’ 이벤트 프로시저가 추가됩니다.

11 현재 통합 문서의 Open 이벤트 프로시저 작성하기

파일이 열릴 때 폼이 실행되도록 frm로그인.Show를 입력합니다.12 

2

1 더블클릭

입력

[프로시저]에서 [BeforClose]를 

클릭합니다. ‘Workbook_BeforClose’ 이벤트 프로시저가 추가됩니다.
13 현재 통합 문서의 BeforClose 이벤트 프로시저 작성하기

파일이 닫히기 전에 [시작화면] 시트를 다시 표시하고 다른 시트는 모두 숨기기하는 코드를 입

력합니다. 이때 시트를 숨기기 위해 사용하는 Visible 속성값으로 xlSheetVeryHidden을 사
용해야 시트 탭을 마우스 오른쪽 버튼으로 클릭하여 [숨기기 취소]를 실행할 수 없습니다.

14 

1    Private Sub Workbook_BeforeClose(Cancel As Boolean)

2    Dim 시트 As Worksheet

3        Sheets("시작화면").Visible = xlSheetVisible

4        For Each 시트 In Worksheets

5            If 시트.Name <> "시작화면" Then 시트.Visible = xlSheetVeryHidden

6        Next

7        ThisWorkbook.Save

8    End Sub

입력

1   통합 문서가 닫히기 직전에 실행되는 이벤트 프로시저입니다.

2   반복문에 사용할 ‘시트’ 변수를 선언합니다.

3   [시작화면] 시트를 숨기기 취소합니다. 다른 시트를 숨기기 전에 최소 한 개의 시트는 숨기기 취소가 되

어 있어야 하므로 [시작화면] 시트의 숨기기 취소가 먼저 실행되도록 합니다.

4   반복문을 시작합니다. 현재 통합 문서의 모든 시트를 ‘시트’ 변수에 저장한 후 시트 수만큼 5를 반복 실

행합니다.

5   시트 이름이 ‘시작화면’이 아니면 숨기기 취소하는 조건문입니다. Visible 속성값을 xlSheetVery 

Hidden으로 설정하면 워크시트에서 숨기기 취소를 할 수 없습니다. 한 줄로 입력한 If 조건문으로, 

End If가 없습니다.

6   반복문을 종료합니다.

7   현재 통합 문서를 저장합니다.

실력향상 워크시트 개체의 Visible 속성에 사용되는 상수에는 xlSheetVisible, xlSheetHidden, xlSheetVeryHidden이 있습

니다. xlSheetVisible은 시트를 표시합니다. xlSheetHidden과 xlSheetVeryHidden은 시트를 숨기기하는데, xlSheetHidden은 워

크시트에서 숨기기 취소를 할 수 있고, xlSheetVeryHidden은 숨기기 취소를 할 수 없습니다. xlSheetVeryHidden으로 지정하면 

시트 탭을 마우스 오른쪽 버튼으로 클릭했을 때 [숨기기 취소] 목록에 시트 이름이 표시되지 않습니다. VBA로 다시 Visible 속성을 

xlSheetVisible로 설정해야 화면에 표시됩니다.

1  [파일] 탭-[다른 이름으로 저장]을 클릭하여 2  파일 형식에서 [Excel 매크

로 사용 통합 문서(*.xlsm)]을 클릭한 후 3  [저장]을 클릭합니다. 저장한 파일을 닫고 다시 열면 매

크로가 실행되지 않아 [시작화면] 시트만 표시됩니다. 4  [콘텐츠 사용]을 클릭합니다.

15 결과 확인하기 

1

2
3 4

1  [아이디]에 onwings, [비밀번호]에 1234를 입력하고 2  [로그인]을 클릭합니다. 로그인 정

보가 확인되었다는 메시지가 나타나면 3  [확인]을 클릭합니다. [시작화면] 시트는 숨기기되고, [인

사기록부] 시트와 [로그인정보] 시트만 표시됩니다.

16  

파일을 닫고 다시 엽니다. 1  이번에는 아이디와 비밀번호를 임의로 입력한 후 2  [로그인]을 클

릭합니다. 로그인 정보가 일치하지 않는다는 메시지가 나타나면 3  [확인]을 클릭합니다. [시작화

면] 시트만 표시됩니다. 

17  

실력향상 다른 이름으로 저장한 매크로 통합 문서를 처음 열었을 때는 보안 경고가 표시되지만 같은 파일을 다시 열면 신뢰할 수 

있는 문서로 등록되어 보안 경고 메시지가 표시되지 않습니다.

1 입력

2

3

2

3
1 임의로 입력

[Visual Basic 편집기]에서 작성한 사용자 정의 폼이나 모듈을 다른 사람이 확인하거나 변경할 수 없도록 VBA 프로젝

트 암호를 설정할 수 있습니다.

비법
노트 VBA 프로젝트 암호 설정하기

[Visual Basic 편집기]에서 [도구] - [VBAProject 속성]을 클릭합니다. 01

[프로젝트 속성] 대화상자의 1  [보호] 탭을 클릭하고 [프로젝트 잠금]에서 2  [읽기 전용으로 프로젝트 잠금]에 체

크합니다. [프로젝트 속성 보기 암호]에서 3  [암호]와 [암호 확인]을 동일하게 입력한 후 4  [확인]을 클릭합니다.  

02

VBA 프로젝트가 보호되었는지 확인하기 위해 현재 통합 문서를 닫은 후 다시 엽니다. [Visual Basic 편집기]의 

[프로젝트] 탐색기 창에서 파일을 클릭하면 암호를 입력하는 대화상자가 나타납니다. 

03

3 입력

1

2

4

사용자 정의 폼, 콤보 상자, 목록 상자, 텍스트 상자, 이미지, 확인란, 옵션 단추, 명령 단추, Initialize 이벤트,

ListIndex 속성, SetFocus 메서드, SendKeys 메서드, GetOpenFilename 메서드, Show 메서드,

Picture 속성, EntireRow 속성, Find 메서드, LoadPicture 함수, Date 함수, Unload 문, If 조건문,

Do While 반복문

리모델링 수주 고객관리
프로그램 만들기

실습 파일 | 별책부록/리모델링고객관리.xlsx   

완성 파일 | 별책부록/리모델링고객관리(완성).xlsm

회사에서
바로 통하는
키워드

02
고급 실무 프로젝트

01   프로젝트 시작하기

주택, 아파트의 리모델링을 의뢰하는 고객의 정보를 등록하고, 수정 또는 삭제할 수 있는 고객관리 프로

그램을 만들어보겠습니다. 

사용자 정의 폼을 이용하여 [고객등록] 폼과 [고객 검색/삭제] 폼을 삽입한 후 [고객등록] 폼에서는 신규 

고객의 정보와 리모델링 항목, 비용, 사진 등을 등록하여 [고객DB] 시트에 자동 입력되도록 합니다. [고

객 검색/삭제] 폼에서는 검색 단어를 입력하면 [고객DB] 시트에서 데이터를 검색하여 폼에 표시하고, 수

정사항이 있을 경우 폼에서 수정하면 [고객DB] 시트에도 자동 수정되도록 프로그램을 작성해보겠습니다.

02   프로젝트 예제 미리 보기

한눈에 
보는 
작업순서

고객등록
사용자 정의 폼
만들기

고객 상세 정보 표시
이벤트 프로시저
작성하기

고객 검색/삭제
사용자 정의 폼
만들기

고객등록 폼의
각 컨트롤 이벤트 프로시저
작성하기

고객 정보
검색하는 프로시저
작성하기

고객 정보 수정과 삭제
이벤트 프로시저
작성하기

메뉴 시트
만들기

03   핵심기능 미리 보기

STEP 02 고객등록 폼 이벤트 프로시저 작성하기

1   ‘Initialize’ 이벤트 프로시저를 추가하여 콤보 상자의 목록을 표시하고 ‘txt계약일자’ 텍스트 상

자에 오늘 날짜를 표시합니다.

2   [사진 찾기] 단추의 ‘Click’ 이벤트 프로시저를 추가하여 고객 사진을 선택하면 이미지 컨트롤에 

표시하고 사진 파일 이름을 ‘txt사진파일명’ 상자에 표시합니다.

3   [신규 등록] 단추의 ‘Click’ 이벤트 프로시저를 추가하여 폼에 입력한 내용이 [고객DB] 시트에 

자동 입력되도록 합니다.

4   [입력 초기화] 단추의 ‘Click’ 이벤트 프로시저를 추가하여 폼의 컨트롤 값이 모두 초기화되도록 

합니다.

STEP 01 고객등록과 검색 사용자 정의 폼 작성하기

1  사용자 정의 폼을 두 개 삽입한 후 컨트롤들을 추가하여 [고객등록] 폼을 작성합니다.

2  [고객 검색/삭제] 폼을 작성합니다.

3  각 폼의 탭 순서를 지정합니다.

STEP 03 고객 검색/삭제 폼 이벤트 프로시저 작성하기

1   [검색하기]의 ‘Click’ 이벤트 프로시저를 추가하여 성명과 전화번호로 [고객DB] 시트에서 고객 

정보를 검색하여 ‘lst검색’ 목록 상자에 표시합니다.

2   ‘lst검색’ 목록 상자의 ‘Click’ 이벤트 프로시저를 추가하여 선택한 고객의 상세 정보가 [고객 검

색/삭제] 폼에 표시되도록 합니다. 

3   [수정 사항 등록]의 ‘Click’ 이벤트 프로시저를 추가하여 수정한 고객 정보가 [고객DB] 시트에 

입력되도록 합니다.

4   [고객 삭제]의 ‘Click’ 이벤트 프로시저를 추가하여 선택한 고객 정보를 [고객DB] 시트에서 삭제

합니다.

5   [고객등록] 폼과 [고객 검색/삭제] 폼을 실행하는 프로시저를 공용 모듈에 작성하고 [메뉴] 시트

에 단추(양식 컨트롤)를 추가하여 프로시저를 연결합니다.

고객등록과 고객 검색/삭제 사용자 정의 폼 작성하기01
STEP

사용자 정의 폼을 두 개 삽입하여 신규 고객을 등록하는 폼과 [고객DB] 시트에 저장된 고객 정보를 검

색하고 수정/삭제하는 폼으로 설정한 후 탭 순서를 지정해보겠습니다.

실습 파일을 엽니다. [Visual Basic 편집기]에서 1  [삽입]-[사용자 

정의 폼]을 두 번 클릭합니다. 사용자 정의 폼이 두 개 추가됩니다. 2  [UserForm1]을 클릭한 후 

폼의 오른쪽 하단에 있는 크기 조절점을 드래그하여 크기를 변경합니다. 3  [속성] 창에서 [이름]에 

frm등록, [Caption]에 고객등록을 입력합니다.

01 사용자 정의 폼 추가하기 

[고객등록] 폼에 다음과 같이 컨트롤을 추가하

고 각 컨트롤의 속성을 변경합니다. 
02 고객등록 폼에 컨트롤 추가하고 속성 설정하기 

실력향상

프레임 안에 있는 컨트롤은 프

레임을 먼저 추가한 후에 추가

해야 합니다.

3 입력

1 2회 클릭

2 드래그

1

2

3
4

5

1. ‘기본사항’ 프레임 

2. ‘고객정보’ 프레임 

번호 컨트롤 종류 속성

1 프레임 •Caption : 기본사항

2 레이블 3개 •Caption : 주택유형, 계약일자, 공사일자

3 콤보 상자 •이름 : cbo유형

4 텍스트 상자 2개
•이름 : txt계약일자
•이름 : txt공사일자

번호 컨트롤 종류 속성

1 프레임 •Caption : 고객정보

2 레이블 7개
•Caption : 성 명, 휴대폰번호, (* 하이픈 포함 입력), 주소, email, 특기사항, 고객사진
•괄호가 있는 레이블은 Font 변경 : 크기(8)

3 텍스트 상자 6개
•이름 : txt성명, txt번호, txt주소, txt메일, txt특기, txt사진파일명
•Font(txt사진파일명) : 글꼴 크기(8)
•Locked(txt사진파일명) : True

4 명령 단추
•이름 : cmd사진찾기
•Caption : 사진 찾기

5 이미지
•이름 : img사진
•PictureSizeMode : 3-frmPictureModeZoom

1

1

2
3

3 3
3
3
3

3

4

4
5

4

3. ‘리모델링 항목’ 프레임 

번호 컨트롤 종류 속성

1 프레임 •Caption : 리모델링 항목

2 확인란 7개
•이름 : chk거실, chk주방, chk안방, chk작은방, chk욕실, chk현관, chk발코니
•Caption : 거실, 주방, 안방, 작은방, 욕실, 현관, 발코니

1

2

2

2

4. ‘금액’ 프레임 

1
3

4

번호 컨트롤 종류 속성

1 프레임 •Caption : 금액

2 레이블 2개 •Caption : 공사비, 결제방법

3 텍스트 상자 •이름 : txt공사비

4 옵션 단추 2개
•이름 : opt현금, opt카드
•Caption : 현금, 카드

5.명령 단추

컨트롤 종류 속성

명령 단추 5개
•이름 : cmd등록, cmd닫기, cmd초기화, cmd고객명단, cmd검색
•Caption : 신규등록, 닫기, 입력초기화, 고객DB조회, 고객검색

1  폼의 빈 공간을 마우스 오른쪽 버튼으로 클릭한 후 2  [탭 순서]를 클릭

합니다. 3  [탭 순서] 대화상자에서 [위로 이동]과 [아래로 이동]을 클릭하여 [Frame1] → [Frame2] 

→ [Frame3] → [Frame4] → [cmd등록] → [cmd닫기] → [cmd초기화] → [cmd고객명단] → 

[cmd검색] 순서로 설정합니다. 4  [확인]을 클릭합니다.

03 탭 순서 지정하기 

실력향상 프레임 안에 있는 컨트롤들은 각 프레임마다 따로 탭 순서를 지정해야 합니다.

2

1 마우스 오른쪽 버튼 클릭

2

4

3

1  ‘기본사항’ 프레임의 빈 공간을 마우스 오른쪽 버튼으로 클릭한 후 2  [탭 순서]를 클릭합니

다. 3  [탭 순서] 대화상자에서 [위로 이동]과 [아래로 이동]을 클릭하여 [cbo유형] → [txt계약일자] 

→ [txt공사일자] 순서로 설정합니다. 레이블은 탭으로 이동할 필요가 없으므로 [txt공사일자] 뒤로  

이동합니다. 4  [확인]을 클릭합니다.

04  

1  ‘고객정보’ 프레임의 빈 공간을 마우스 오른쪽 버튼으로 클릭한 후 2  [탭 순서]를 클릭합

니다. 3  [탭 순서] 대화상자에서 [위로 이동]과 [아래로 이동]을 클릭하여 [txt성명] → [txt번호] → 

[txt주소] → [txt메일] → [txt특기] → [cmd사진찾기] → [txt사진파일명] 순서로 설정합니다. 나머

지 컨트롤은 탭으로 이동할 필요가 없으므로 [txt사진파일명] 뒤로 이동합니다. 4  [확인]을 클릭합

니다.

05 

1 마우스 오른쪽 버튼 클릭

2

4

3

1 마우스 오른쪽 버튼 클릭

2

4

3

1  ‘리모델링 항목’ 프레임의 빈 공간을 마우스 오른쪽 버튼으로 클릭한 후 2  [탭 순서]를 클

릭합니다. 3  [탭 순서] 대화상자에서 [위로 이동]과 [아래로 이동]을 클릭하여 [chk거실] → [chk주

방] → [chk안방] → [chk작은방] → [chk욕실] → [chk현관] → [chk발코니] 순서로 설정합니다. 

4  [확인]을 클릭합니다.

06 

1  ‘금액’ 프레임의 빈 공간을 마우스 오른쪽 버튼으로 클릭한 후 2  [탭 순서]를 클릭합니다. 

3  [탭 순서] 대화상자에서 [위로 이동]과 [아래로 이동]을 클릭하여 [txt공사비] → [opt현금] → 

[opt카드] 순서로 설정합니다. 4  [확인]을 클릭합니다.

07 

1 마우스 오른쪽 버튼 클릭

1 마우스 오른쪽 버튼 클릭

2

4

3

2

4

3

1  [UserForm2]를 클릭한 후 폼의 오른쪽 하단에 있는 크기 조

절점을 드래그하여 크기를 변경합니다. 2  [속성] 창에서 [이름]에 frm검색, [Caption]에 고객 검
색/삭제를 입력합니다.

08 검색/삭제 폼 속성 변경하기 

[고객 검색/삭제] 폼에 다음과 같이 컨

트롤을 추가하고 각 컨트롤의 속성을 변경합니다. 
09 고객 검색/삭제 폼에 컨트롤 추가하고 속성 설정하기 

2 입력

1 드래그

1

2

3

1.검색어 입력 부분

번호 컨트롤 종류 속성

1 레이블 4개
•Caption : 검색 성명, 휴대폰번호, (1글자 이상 입력), (뒷자리 4글자 입력)
•괄호가 있는 레이블은 Font 변경 : 크기(8)

2 텍스트 상자 2개 •이름 : txt검색성명, txt검색번호

3 명령 단추
•이름 : cmd검색
•Caption : 검색하기

번호 컨트롤 종류 속성

1 프레임 • Caption : 고객상세정보

2 레이블 12개
• Caption : 성 명, 주택유형, 계약일자, 공사
일자, 휴대폰번호, 공사비, 결제방법, 주소, 
email, 특기 사항, 리모델링 항목, 사진

3 텍스트 상자 10개

• 이름 : txt조회성명, txt조회유형, txt조회계약
일자, txt조회공사일자, txt조회번호, txt조회
공사비, txt조회결제방법, txt조회주소, txt조
회메일, txt조회특기

4 목록 상자 •이름 : lst조회항목

5 이미지 •이름 : img조회사진

6 명령 단추 2개
•이름 : cmd수정, cmd삭제
•Caption : 수정 사항 등록, 고객 삭제

번호 컨트롤 종류 속성

1 레이블 •Caption : 성   명       |   주택 유형    |  휴대폰 번호

2 목록 상자

•이름 : lst검색
•ColumnCount : 4
•ColumnWidths : 40pt; 60pt; 70pt; 0pt
•ListStyle : 1-fmListStyleOption

3 명령 단추 3개
•이름 : cmd등록, cmd고객DB, cmd닫기
•Caption : 신규 등록, 고객 DB, 닫 기

2 2 3

2.검색 결과 표시 부분

1

2

3

3. ‘고객상세정보’ 프레임
1

3

4

5

6

1

2

2



030

1  ‘고객상세정보’ 프레임을 제외한 빈 공간을 마우스 오른쪽 버튼으로 클릭

한 후 2  [탭 순서]를 클릭합니다. 3  [탭 순서] 대화상자에서 [위로 이동]과 [아래로 이동]을 클릭하

여 [txt검색성명] → [txt검색번호] → [lst검색] → [cmd검색] → [Frame1] → [cmd등록] → [cmd

고객DB] → [cmd닫기] 순서로 설정합니다. 4  [확인]을 클릭합니다.

10 탭 순서 지정하기 

1  ‘고객상세정보’ 프레임의 빈 공간을 마우스 오른쪽 버튼으로 클릭한 후 2  [탭 순서]를 클릭

합니다. 3  [탭 순서] 대화상자에서 [위로 이동]과 [아래로 이동]을 클릭하여 [txt조회성명] → [txt

조회유형] → [txt조회계약일자] → [txt조회공사일자] → [txt조회번호] → [txt조회공사비] → [txt

조회결제방법] → [txt조회주소] → [txt조회메일] → [txt조회특기] → [txt조회항목] → [cmd수정] 

→ [cmd삭제] 순서로 설정합니다. 4  [확인]을 클릭합니다.

11  

1 마우스 오른쪽 버튼 클릭

2

4

3

1 마우스 오른쪽 버튼 클릭

2

4

3

고객등록 폼 이벤트 프로시저 작성하기 02
STEP

[고객등록] 폼이 실행되면서 준비해야 할 내용을 ‘Initialize’ 이벤트 프로시저에 작성하고, [사진 찾기] 

단추에는 [사진] 폴더 안에 준비된 고객 사진을 찾아오도록 이벤트 프로시저를 작성합니다. [신규 등록] 

단추에는 폼에 입력된 항목들이 [고객DB] 시트에 자동 입력되도록 하는 이벤트 프로시저를, [입력 초기

화] 단추에는 폼의 컨트롤들을 모두 초기화하는 이벤트 프로시저를 작성해보겠습니다.  

1  [프로젝트] 탐색기 창에서 [폼]-[frm검색]을 더블클

릭합니다. [고객등록] 폼이 나타나면 2  폼의 빈 공간을 더블클릭합니다. [코드] 창이 표시되면 3  

[프로시저]에서 [Initialize]를 선택합니다.

12 Initialize 이벤트 프로시저 작성하기 

실력향상 Initialize 이벤트는 폼이 실행되어 화면에 보여지기 바로 전에 한 번 발생하는 이벤트입니다. 

‘UserForm_Initialize’ 이벤트 프로시저가 생성됩니다. 1  ‘UserForm_Click’ 이벤트 프로시

저는 범위로 지정하고 2  Delete 를 눌러 삭제합니다. 3  ‘UserForm_Initialize’ 이벤트 프로시저에 

주택유형 항목을 추가하고 계약 날짜로 오늘 날짜를 표시하는 코드를 작성합니다.

13  

1 더블클릭

2 더블클릭

3

2 Delete 3 입력1 드래그

1    Private Sub UserForm_Initialize()

2        With Me.cbo유형

3            .AddItem "아파트"

4            .AddItem "단독주택"

5            .AddItem "연립주택"

6            .AddItem "다가구주택"

7        End With

9        Me.txt계약일자.Value = Date    

End Sub

1   현재 폼이 실행되기 바로 전에 발생하는 ‘Initialize’ 이벤트 프로시저를 시작합니다.

2   Me.cbo유형을 With 문으로 묶어서 3  ~ 6의 마침표(.) 앞에 생략되었음을 표시합니다. 

3   ~ 6  ‘cbo유형’ 콤보 상자에 항목을 하나씩 추가합니다. AddItem 메서드는 콤보 상자에 행 목록을 추

가 합니다. 

7   With 문을 종료합니다.

8   ‘txt계약일자’ 텍스트 상자에 오늘 날짜를 입력합니다. VBA에서 Date 함수는 오늘 날짜를 표시합니다.

1  [Visual Basic 편집기]의 [표준] 도구 모음에서 [Sub/사용자 정의 폼 실행 ]

을 클릭합니다. 폼이 실행되면 2  ‘cbo유형’ 콤보 상자의 목록 단추를 클릭합니다. 추가한 항목이 

표시됩니다. ‘txt계약일자’ 텍스트 상자에는 오늘 날짜가 표시됩니다.

14 결과 확인하기 

1 2

‘고객정보’ 프레임에 있는 ‘cmd사진찾기’ 

명령 단추를 더블클릭합니다. ‘cmd사진찾기_Click’ 이벤트 프로시저가 추가됩니다.
15 사진 찾기 단추의 Click 이벤트 프로시저 작성하기

‘cmd사진찾기_Click’ 이벤트 프로시저에 사진 파일을 선택할 수 있는 열기 대화상자를 표시

한 후 사용자가 선택하는 사진을 이미지 컨트롤에 표시하고 사진 파일 이름은 텍스트 상자에 표시

하도록 하는 코드를 입력합니다.

16  

1    Private Sub cmd사진찾기_Click()

2      Dim 사진파일 As Variant        

3     사진파일 = Application.GetOpenFilename(Filefilter:="그림파일,*.gif;*.jpg;*.bmp", 

          Title:="고객 사진 선택")

4      If 사진파일 = False Then Exit Sub    

5      Me.img사진.Picture = LoadPicture(사진파일)

6      Me.txt사진파일명 = Dir(사진파일)    

End Sub

입력

1   ‘cmd사진 찾기’ 명령 단추를 클릭했을 때 발생할 이벤트 프로시저를 시작합니다.

2   ‘사진파일’ 변수를 가변형으로 선언합니다. 이 변수에는 열기 대화상자에서 선택하는 사진 파일의 정보

를 저장합니다. 열기 대화상자에서 사진을 선택하면 선택한 사진의 경로를 포함한 파일명을 저장하고, 

만약 열기 대화상자에서 [취소]를 클릭하면 변수에 False를 저장하므로 변수를 선언할 때 자료 형식을 

가변형으로 선언합니다.

3   한 줄로 입력합니다. 사진 파일을 선택할 수 있는 열기 대화상자를 표시합니다. 열기 대화상자의 이름은 

‘고객 사진 선택’으로 표시하고, gif, jpg, bmp 파일 형식을 선택할 수 있도록 설정됩니다. 사용자가 선

택하는 사진의 파일 정보를 ‘사진파일’ 변수에 저장합니다.

4   ‘사진파일’ 변수가 False일 경우 프로시저를 종료하는 조건문입니다. 한 줄로 입력하는 If 문으로, End 

If가 없습니다. 만약 열기 대화상자에서 [취소]를 클릭하면 ‘사진파일’ 변수에 False가 저장되어 5에서 

오류가 발생하므로 [취소]를 클릭하면 프로시저를 종료합니다.

5   선택한 사진 파일을 이미지 컨트롤에 표시합니다.

6   선택한 사진 파일의 이름을 ‘txt사진파일명’ 텍스트 상자에 대입합니다. 이 텍스트 상자에 입력된 파일 

이름은 [고객DB] 시트에 저장할 때 사용합니다.

1  [Visual Basic 편집기]의 [표준] 도구 모음에서 [Sub/사용자 정의 폼 실행 ]

을 클릭합니다. 폼이 실행되면 2  ‘cmd사진찾기’ 명령 단추를 클릭합니다. 
17 결과 확인하기 

1

2

1  [사진] 폴더 안에 준비된 사진 중 하나를 클릭한 후 2  [열기]를 클릭합니다. 이미지 컨트롤

에 사진이 표시되고, ‘txt사진파일명’ 텍스트 상자에 사진 파일 이름이 표시됩니다.
18  

‘cmd등록’ 명령 단추를 더블클릭합니다. 

‘cmd등록_Click’ 이벤트 프로시저가 추가됩니다.
19 신규 등록 단추의 Click 이벤트 프로시저 작성하기 

1

2

더블클릭

[고객등록] 대화상자에서 [신규 등록]을 클릭하면 입력한 값들이 [고객DB] 시트에 입력되도록 

하는 코드를 입력합니다.
20  

1    Private Sub cmd등록_Click()

2    Dim 행번호 As Integer

3    행번호 = Sheets("고객DB").Range("A1").CurrentRegion.Rows.Count + 1    

4        With Sheets("고객DB")

5            .Cells(행번호, 1) = Format(Me.txt계약일자, "mmdd") & "-" & Right(Me.txt번호, 4)

6            .Cells(행번호, 2) = Me.cbo유형

7            .Cells(행번호, 3) = Me.txt계약일자

8            .Cells(행번호, 4) = Me.txt공사일자

9            .Cells(행번호, 5) = Me.txt성명

10           .Cells(행번호, 6) = Me.txt번호

11            .Cells(행번호, 7) = Me.txt주소

12            .Cells(행번호, 8) = Me.txt메일

13            .Cells(행번호, 9) = Me.txt특기

14            .Cells(행번호, 10) = Me.txt사진파일명  

15            If Me.chk거실 Then .Cells(행번호, 11).Value = "●"

16            If Me.chk주방 Then .Cells(행번호, 12).Value = "●"

17            If Me.chk안방 Then .Cells(행번호, 13).Value = "●"

18            If Me.chk작은방 Then .Cells(행번호, 14).Value = "●"

입력

19            If Me.chk욕실 Then .Cells(행번호, 15).Value = "●"

20           If Me.chk현관 Then .Cells(행번호, 16).Value = "●"

21            If Me.chk발코니 Then .Cells(행번호, 17).Value = "●"     

22          .Cells(행번호, 18) = Val(Me.txt공사비)     

23          If Me.opt현금 Then

24               .Cells(행번호, 19) = "현금"

25           ElseIf Me.opt카드 Then

26               .Cells(행번호, 19) = "카드"

27           End If

28       End With

29       MsgBox Me.txt성명 & "님 리모델링 정보가 등록되었습니다"

End Sub

1   ‘cmd등록’ 명령 단추를 클릭했을 때 실행할 이벤트 프로시저를 시작합니다.

2   [고객DB] 시트에 입력할 셀 위치의 행 번호를 저장할 ‘행번호’ 변수를 선언합니다.

3   [고객DB] 시트의 [A1] 셀을 기준으로 데이터가 입력된 행 개수를 계산하고 그 결과에 1을 더한 값을 ‘행

번호’ 변수에 대입합니다. 

4   Sheets(“고객DB”)를 With 문으로 묶어 5  ~ 26의 마침표(.) 앞에 생략되었음을 표시합니다.

5   A열에 고객코드를 ‘계약일자의 월일-휴대폰번호 뒷자리 4글자’ 형식으로 만들어 입력합니다. 

6   ~ 14 [B:J] 열에 주택유형, 계약일자, 공사일자, 성명, 휴대폰번호, 주소, email, 특기사항, 고객 사진 

파일 이름을 순서대로 입력합니다.

15   ~ 21  [K:Q] 열에 리모델링 항목의 확인란이 선택되어 있으면 각 열에 ●를 입력합니다. 한 줄로 입력하

는 If 조건문으로 End If가 없습니다.

22  R열에 공사비를 숫자 형식으로 변환하여 입력합니다.

23  ~ 27 S열에 결제 방법을 입력하는 조건문입니다. 현금이 선택되어 있으면 현금, 카드가 선택되어 있으

면 카드를 입력합니다.

28  With 문을 종료합니다.

29  등록이 완료되었다는 메시지 대화상자를 나타냅니다.

[개체]에서 [cmd초기화]를 클릭합니다. 

‘cmd초기화_Click’ 이벤트 프로시저가 추가됩니다.
21 입력 초기화 단추의 Click 이벤트 프로시저 작성하기 

[고객등록] 대화상자에서 [입력 초기화]를 클릭하면 입력하거나 선택한 값이 모두 초기화되는 

코드를 입력합니다.
22 

1    Private Sub cmd초기화_Click()

2        Me.cbo유형.ListIndex = -1

3        Me.txt계약일자 = Date

4        Me.txt공사일자 = ""

5        Me.txt성명 = ""

6        Me.txt번호 = ""

7        Me.txt주소 = ""

8        Me.txt메일 = ""

입력

9        Me.txt특기 = ""

10       Me.img사진.Picture = Nothing

11        Me.txt사진파일명 = ""

12        Me.chk거실 = False

13        Me.chk주방 = False

14        Me.chk안방 = False

15        Me.chk작은방 = False

16        Me.chk욕실 = False

17        Me.chk현관 = False

18        Me.chk발코니 = False

19        Me.txt공사비 = ""

20       Me.opt현금 = False

21        Me.opt카드 = False

End Sub

1   ‘cmd초기화’ 명령 단추를 클릭했을 때 실행되는 이벤트 프로시저를 시작합니다.

2   ‘cbo유형’ 콤보 상자에 선택된 내용을 지웁니다. 콤보 상자의 ListIndex 값을 -1로 설정하면 선택된 항

목이 없는 것을 뜻합니다. 

3   ‘txt계약일자’ 텍스트 상자에 오늘 날짜를 입력합니다. 계약일자는 폼이 실행되기 전 Initialize 이벤트

에 의해 오늘 날짜가 표시되므로 초기화 작업으로 오늘 날짜를 입력합니다.

4   ~ 9  ‘txt공사일자’, ‘txt성명’, ‘txt번호’, ‘txt주소’, ‘txt이메일’, ‘txt특기’ 텍스트 상자에 빈 값을 대입합

니다.

10  사진을 지웁니다.

11   ‘txt사진파일명’ 텍스트 상자에 빈 값을 대입합니다.

12   ~ 18  ‘리모델링 항목’을 모두 선택 해제합니다.

19   ‘txt공사비’ 텍스트 상자에 빈 값을 대입합니다.

20  ~ 21  ‘결제방법’ 항목인 ‘opt현금’과 ‘opt카드’의 선택을 모두 해제합니다.

같은 방법으로 [개체]

에서 [cmd닫기], [cmd고객명단], [cmd검색]을 각각 클릭하여 ‘Click’ 이벤트 프로시저를 작성합

니다.

23 닫기, 고객DB조회, 고객 검색 단추의 Click 이벤트 프로시저 작성하기 

1    Private Sub cmd검색_Click()

2        Unload Me

3        frm검색.Show vbModeless

End Sub

1    Private Sub cmd고객명단_Click()

2        Sheets("고객DB").Select

3        Unload Me

End Sub

1   ‘cmd검색’ 명령 단추를 클릭했을 때 실행되는 이벤트 프로시저입니다.

2   현재 폼을 닫습니다. 

3   검색 폼을 모덜리스 형식으로 실행합니다.

1   ‘cmd고객명단’ 명령 단추를 클릭했을 때 실행되는 이벤트 프로시저입니다.

2   [고객DB] 시트를 선택합니다. 

3   현재 폼을 닫습니다.

입력

1    Private Sub cmd닫기_Click()

2        Unload Me

End Sub

1   ‘cmd닫기’ 명령 단추를 클릭했을 때 실행되는 이벤트 프로시저입니다.

2   현재 폼을 닫습니다.

고객 정보를 등록할 때 주택유형, 성명, 휴대폰번호, 리모델링 항목

은 필수 항목으로 꼭 입력하도록 ‘cmd등록’ 명령 단추의 ‘Click’ 이벤트 프로시저를 수정합니다.
24 필수 항목 입력 확인하기 

Private Sub cmd등록_Click()

Dim 행번호 As Integer

행번호 = Sheets("고객DB").Range("A1").CurrentRegion.Rows.Count + 1    

1       If Me.cbo유형.ListIndex = -1 Then

2            MsgBox "주택유형이 선택되지 않았습니다"

3            cbo유형.SetFocus

4            SendKeys "{F4}"

5            Exit Sub

6        End If    

7        If Me.txt성명 = "" Then

8            MsgBox "성명이 입력되지 않았습니다"

입력

9            Me.txt성명.SetFocus

10           Exit Sub

11        End If    

12        If Me.txt번호 = "" Then

13            MsgBox "휴대폰 번호가 입력되지 않았습니다"

14            Me.txt번호.SetFocus

15            Exit Sub

16        End If    

17        If Me.chk거실 = False And Me.chk주방 = False And Me.chk안방 = False And Me.chk 

                작은방 = False And Me.chk욕실 = False And Me.chk현관 = False And Me.chk 

                      발코니 = False Then

18            MsgBox " 리모델링 항목이 선택되지 않았습니다"

19            Exit Sub

20       End If    

        With Sheets("고객DB")

            .Cells(행번호, 1) = Format(Me.txt계약일자, "mmdd") & "-" & Right(Me.txt번호, 4)

            .Cells(행번호, 2) = Me.cbo유형

            .Cells(행번호, 3) = Me.txt계약일자

            .Cells(행번호, 4) = Me.txt공사일자

            .Cells(행번호, 5) = Me.txt성명

            .Cells(행번호, 6) = Me.txt번호

            .Cells(행번호, 7) = Me.txt주소

            .Cells(행번호, 8) = Me.txt메일

            .Cells(행번호, 9) = Me.txt특기

            .Cells(행번호, 10) = Me.txt사진파일명  

            If Me.chk거실 Then .Cells(행번호, 11).Value = "●"

            If Me.chk주방 Then .Cells(행번호, 12).Value = "●"

            If Me.chk안방 Then .Cells(행번호, 13).Value = "●"

            If Me.chk작은방 Then .Cells(행번호, 14).Value = "●"

            If Me.chk욕실 Then .Cells(행번호, 15).Value = "●"

            If Me.chk현관 Then .Cells(행번호, 16).Value = "●"

            If Me.chk발코니 Then .Cells(행번호, 17).Value = "●"     

            .Cells(행번호, 18) = Val(Me.txt공사비)     

            If Me.opt현금 Then

                   .Cells(행번호, 19) = "현금"

            ElseIf Me.opt카드 Then

                   .Cells(행번호, 19) = "카드"

                End If

        End With

1   ~ 6  주택유형을 선택하지 않으면 메시지 대화상자를 나타내고, ‘cho유형’ 콤보 상자의 목록을 펼쳐서 

표시한 후 프로시저를 종료합니다. SendKeys “{F4}”는 키보드의 F4 를 누르는 것과 같습니다.

7   ~ 11  성명을 입력하지 않으면 메시지 대화상자를 나타내고, ‘txt성명’ 텍스트 상자로 커서를 이동시킨 

후 프로시저를 종료합니다. 

12   ~ 16 휴대폰번호를 입력하지 않으면 메시지 대화상자를 나타내고, ‘txt번호’ 텍스트 상자로 커서를 이동

시킨 후 프로시저를 종료합니다.

17   ~ 20 ‘리모델링 항목’ 프레임에 있는 확인란은 최소 한 개 이상 선택되어야 하는데 모두 선택되지 않으면 

메시지 대화상자를 나타내고 프로시저를 종료합니다. 17은 한 줄로 입력합니다.

21   ‘cmd초기화_Click’ 이벤트 프로시저를 호출하여 실행합니다. [고객DB] 시트에 내용을 모두 입력한 후 

컨트롤들을 초기화합니다.

1  [Visual Basic 편집기]의 [표준] 도구 모음에서 [Sub/사용자 정의 폼 실행

]을 클릭합니다. 폼이 실행되면 2  각 컨트롤에 내용을 입력하거나 선택한 후 [신규 등록]을 클릭

합니다. 등록되었다는 메시지가 나타나면 3  [확인]을 클릭합니다.

25 결과 확인하기 

        MsgBox Me.txt성명 & "님 리모델링 정보가 등록되었습니다"    

21        Call cmd초기화_Click

End Sub

1

3

2

[고객DB] 시트에 입력된 내용을 확인합니다.26  

다시 폼을 실행한 후 아무것도 입력하지 않은 상태에서 1  [신규 등록]을 클릭합니다. 주택유

형이 선택되지 않았다는 메시지가 나타나고 2  [확인]을 클릭하면 [주택유형] 목록이 펼쳐져 표시됩

니다.

27  

2

1

고객 검색/삭제 폼의 이벤트 프로시저 작성하기 03
STEP

[고객 검색/삭제] 폼에서 검색어를 입력한 후 [검색하기]를 클릭하면 [고객DB] 시트에서 고객 정보를 검

색하여 목록 상자에 표시하고, 목록 상자에서 한 명의 고객을 클릭하면 고객상세정보가 표시되도록 이

벤트 프로시저를 작성해보겠습니다. 또한 고객 정보를 수정하거나 삭제하면 [고객DB] 시트에 반영되도

록 이벤트 프로시저를 작성합니다.

1  [프로젝트] 탐색기 창에서 [폼]-[frm검

색]을 더블클릭합니다. 폼이 표시되면 2  ‘cmd검색’ 명령 단추를 더블클릭합니다. [코드] 창이 표시

되고 ‘cmd검색_Click’ 이벤트 프로시저가 추가됩니다.

28 검색하기 단추의 Click 이벤트 프로시저 작성하기 

‘txt검색성명’과 ‘txt검색번호’의 텍스트 상자에 한 글

자 이상 입력한 후 [검색하기]를 클릭하면 [고객DB] 시트에서 일치하는 고객 정보를 찾아 목록 상

자에 표시하는 문장을 입력합니다.

29 성명과 휴대폰번호로 고객 정보 검색하기 

2

1 더블클릭

입력

1    Private Sub cmd검색_Click()

2    Dim 셀위치 As Range

3    Dim 행번호 As Integer

4    Set 셀위치 = Sheets("고객DB").Range("E2")

5    행번호 = 0

6    Me.lst검색.Clear

7    Do While 셀위치.Value <> ""

8        If 셀위치.Value Like "*" & Me.txt검색성명 & "*" And 셀위치.Offset(0, 1).Value 

               Like "*" & Me.txt검색번호 & "*" Then

9            Me.lst검색.AddItem 셀위치.Value

10           Me.lst검색.Column(1, 행번호) = 셀위치.Offset(0, -3)

11            Me.lst검색.Column(2, 행번호) = 셀위치.Offset(0, 1)

12            Me.lst검색.Column(3, 행번호) = 셀위치.Offset(0, -4)

13            행번호 = 행번호 + 1

14        End If

15        Set 셀위치 = 셀위치.Offset(1, 0)

16    Loop

End Sub

1   ‘cmd검색’ 명령 단추를 클릭했을 때 실행되는 이벤트 프로시저를 시작합니다.

2   [고객DB] 시트에서 검색할 성명의 셀 위치를 저장할 ‘셀위치’ 변수를 선언합니다.

3   목록 상자에 추가되는 행 목록 수를 저장할 ‘행번호’ 변수를 선언합니다.

4   ‘셀위치’ 변수에 [고객DB] 시트의 [E2] 셀을 대입합니다. [E2] 셀에는 등록된 고객의 첫 번째 성명이 있

습니다. 

5   ‘행번호’ 변수에 0을 대입합니다. 목록 상자의 ListIndex 시작값이 0이기 때문에 0을 대입합니다.

6   ‘lst검색’ 목록 상자를 초기화하는 문장으로, 목록 상자에 표시된 항목이 있다면 모두 제거합니다.

7   ‘셀위치’ 변수에 저장된 셀에 값이 있는 동안 8  ~ 15를 반복 실행합니다.

8   조건문을 시작하여 조건식이 참이면 9  ~ 13을 실행합니다. ‘txt검색성명’ 텍스트 상자의 값과 [고객

DB] 시트의 성명을 비교하고, ‘txt검색’ 텍스트 상자의 값과 [고객DB] 시트의 휴대폰번호를 비교합니

다. 별표(*)를 이용하여 포함된 문자를 비교할 때는 Like 연산자를 사용합니다. 한 줄로 입력합니다.

9   ‘lst검색’ 목록 상자에 성명을 추가합니다. 첫 번째 열을 추가할 때는 AddItem 메서드를 사용합니다.

10  ~ 12  추가한 성명과 같은 ListIndex에 주택유형, 휴대폰번호, 고객코드를 열 단위로 추가합니다. 두 번

째 열에 항목을 추가할 때는 Column 속성을 사용합니다. 셀위치.Offset(0, -3)는 성명이 입력된 열을 

기준으로 왼쪽으로 세 칸 이동한 셀의 값을 뜻하므로 주택유형이 됩니다. 

13   목록 상자의 목록이 한 개 추가되었으므로 ‘행번호’ 변수를 1 증가시킵니다.

14   If 조건문을 종료합니다.

15   다음 성명을 비교하기 위해 아래쪽으로 한 칸 이동한 셀을 ‘셀위치’ 변수에 대입합니다.

16   반복문을 종료합니다.

1  [Visual Basic 편집기]의 [표준] 도구 모음에서 [Sub/사용자 정의 폼 실행

]을 클릭합니다. 폼이 실행되면 2  [검색 성명]에 김을 입력한 후 3  [검색하기]를 클릭합니다. 조

건에 맞는 고객 정보가 검색되어 목록 상자에 표시됩니다. 4  [닫기 ]를 클릭해 폼을 닫습니다.

30 결과 확인하기 

1  [삽입]-[모듈]을 클릭합니다. 2  모듈이 추

가되면 [속성] 창에서 [이름]을 검색프로그램으로 변경합니다.
31 선택한 고객의 셀 위치를 찾는 프로시저 작성하기 

실력향상 목록 상자에 표시된 고객 중 한 명을 선택했을 때 그 정보가 오른쪽 [고객상세정보] 프레임에 표시되도록 하려면 선택

한 고객의 정보가 [고객DB] 시트의 어디에 입력되어 있는지 알아야 합니다. 이 기능은 고객 정보를 수정하거나 삭제할 때도 필요한 기

능으로, 공용 모듈에 작성합니다. 공용 모듈에 작성한 프로시저는 현재 통합 문서에서 필요할 때마다 Call 문으로 호출하여 사용할 수 

있습니다.

1

2 입력 3 4

1

2 입력

현재 통합 문서에서 모두 사용할 수 있는 프로젝트 수준의 변수를 선언하고 목록 상자에서 

선택한 고객 정보가 [고객DB] 시트의 어디에 입력되어 있는지 셀 위치를 찾아내는 ‘고객정보검색’ 

프로시저를 작성합니다.

32  

1    Public 고객코드 As String

2    Public 고객범위 As Range

3    Public 셀위치 As Range

4    Sub 고객정보검색()

5    On Error GoTo 에러처리

6    고객코드 = frm검색.lst검색.Column(3, frm검색.lst검색.ListIndex)

7    Set 고객범위 = Sheets("고객DB").Range("A2", Sheets("고객DB").Range("A2").End(xlDown))

8    Set 셀위치 = 고객범위.Find(고객코드, Lookat:=xlWhole)

9    Exit Sub

10   에러처리:

11        MsgBox "목록에서 고객을 먼저 선택하세요"    

End Sub

1   목록 상자에서 선택한 고객의 고객코드를 저장할 ‘고객코드’ 변수를 선언합니다. 이 변수가 사용되는 ‘고

객정보검색’ 프로시저는 [고객 검색/삭제] 폼에서 여러 번 호출하여 사용되므로 프로시저 밖에서 Public

으로 선언해야 합니다.

2   [고객DB] 시트에 입력된 고객코드의 범위를 저장할 ‘고객범위’ 변수를 선언합니다.

3   일치하는 고객코드를 찾았을 때 셀 위치를 저장할 ‘셀위치’ 변수를 선언합니다. 목록 상자에서 선택한 고

객을 [고객DB] 시트에서 찾을 때 고객코드를 기준으로 찾아야 중복값이 없습니다.

4   ‘고객정보검색’ 프로시저를 시작합니다.

5   오류가 발생하면 ‘에러처리’ 레이블이 있는 곳으로 이동합니다. 만약 목록 상자에서 고객이 선택되지 않

은 상태에서 이 프로시저가 실행된다면 6에서 오류가 발생하므로 이를 해결하기 위한 코드입니다.

입력

6   목록 상자에서 선택한 고객의 고객코드를 ‘고객코드’ 변수에 대입합니다. 목록 상자의 Column 속성은 

‘Column(열번호, 행번호)’ 형식으로 사용하는데, 열 번호는 네 번째 열이므로 3이고, 행 번호는 목록 상

자에서 선택된 행이 바로 ListIndex이므로 ListIndex 속성값을 행 번호로 사용합니다.

7   [고객DB] 시트의 [A2] 셀에서 Ctrl + ↓를 눌렀을 때 선택되는 범위를 ‘고객범위’ 변수에 대입합니다.

8   Find 메서드를 이용하여 ‘고객범위’ 변수에 저장된 범위에서 ‘고객코드’ 변수에 저장된 코드와 일치하는 

값을 찾아 해당 셀을 ‘셀위치’ 변수에 대입합니다. Find 메서드의 Lookat:=xlWhole 인수를 설정하면 

전체 셀 데이터가 일치하는 항목을 찾습니다. ‘셀위치’ 변수를 프로시저 밖에 Public으로 선언했기 때문

에 프로시저가 종료되더라도 저장된 값은 유지됩니다.

9   프로그램이 정상적으로 실행되었을 때 10 ~ 11이 실행되지 않도록 프로시저를 종료합니다.

10  오류가 발생했을 때 이동할 곳입니다.

11   목록 상자에서 고객을 먼저 선택하도록 메시지 대화상자를 나타냅니다.

1  [프로젝트] 탐색기 창에서 [폼]-[frm검색]을 

더블클릭합니다. 2  폼이 표시되면 ‘lst검색’ 목록 상자를 더블클릭합니다. ‘lst검색_Click’ 이벤트 

프로시저가 추가됩니다.

33 목록 상자의 Click 이벤트 프로시저 작성하기 

2 더블클릭

1 더블클릭

‘lst검색’ 목록 상자에서 고객 한 명을 클릭하면 상세 정보가 ‘고객상세정보’ 프레임의 각 컨트

롤에 표시되도록 하는 코드를 입력합니다.
34 

1    Private Sub lst검색_Click()

2    Call 고객정보검색

3        Me.txt조회성명 = 셀위치.Offset(0, 4)

4        Me.txt조회유형 = 셀위치.Offset(0, 1)

5        Me.txt조회계약일자 = 셀위치.Offset(0, 2)

6        Me.txt조회공사일자 = 셀위치.Offset(0, 3)

7        Me.txt조회번호 = 셀위치.Offset(0, 5)

8        Me.txt조회주소 = 셀위치.Offset(0, 6)

9        Me.txt조회메일 = 셀위치.Offset(0, 7)

10       Me.txt조회특기 = 셀위치.Offset(0, 8)    

11        If 셀위치.Offset(0, 9) <> "" Then Me.img조회사진.Picture = _

           LoadPicture(ThisWorkbook.Path & "\사진\" & 셀위치.Offset(0, 9))    

12        Me.lst조회항목.Clear

13        If 셀위치.Offset(0, 10) = "●" Then Me.lst조회항목.AddItem "거실"

14        If 셀위치.Offset(0, 11) = "●" Then Me.lst조회항목.AddItem "주방"

15        If 셀위치.Offset(0, 12) = "●" Then Me.lst조회항목.AddItem "안방"

16        If 셀위치.Offset(0, 13) = "●" Then Me.lst조회항목.AddItem "작은방"

17        If 셀위치.Offset(0, 14) = "●" Then Me.lst조회항목.AddItem "욕실"

18        If 셀위치.Offset(0, 15) = "●" Then Me.lst조회항목.AddItem "현관"

19        If 셀위치.Offset(0, 16) = "●" Then Me.lst조회항목.AddItem "발코니"        

20       Me.txt조회공사비 = 셀위치.Offset(0, 17)

21        Me.txt조회결제방법 = 셀위치.Offset(0, 18)

End Sub

입력

1   ‘lst검색’ 목록 상자를 클릭했을 때 발생할 이벤트 프로시저를 시작합니다.

2   ‘고객정보검색’ 프로시저를 호출합니다. 이 프로시저는 [검색프로그램] 모듈에 작성되어 있으며 이 프로

시저가 실행되고 나면 ‘셀위치’ 변수에는 목록 상자에서 선택한 고객의 고객코드가 [고객DB] 시트의 어

느 셀에 있는지 셀 정보를 가지고 복귀합니다.

3   ~ 10 목록 상자에서 선택한 고객의 정보를 ‘고객상세정보’ 프레임에 있는 텍스트 상자에 표시합니다. 

‘셀위치’ 변수에는 [고객DB] 시트의 A열 중 한 셀의 정보가 저장되어 있으므로 성명, 주택유형 등의 정보

는 A열을 기준으로 Offest 속성으로 열을 이동하며 해당 항목의 값을 가져옵니다.

11   이미지 컨트롤에 사진 파일을 표시합니다. 사진 파일은 현재 통합 문서가 있는 폴더 안의 [사진] 폴더에 

반드시 저장되어 있어야 합니다. 한 문장인데 언더바(_)를 사용하여 두 줄로 나누어 작성했습니다.

12   ‘리모델링 항목’의 ‘lst조회항목’ 목록 상자에 새로운 정보를 표시하기 위해 기존에 있는 목록을 모두 삭

제합니다. 

13   ~ 19 각 리모델링 항목 열에 ●가 표시되어 있으면 해당 항목을 목록 상자에 추가합니다.

20  공사비를 ‘txt조회공사비’ 텍스트 상자에 표시합니다.

21   결제방법을 ‘txt조회결제방법’ 텍스트 상자에 표시합니다.

[개체]에서 [cmd수정]을 클릭합니

다. ‘cmd수정_Click’ 이벤트 프로시저가 추가됩니다.
35 수정 사항 등록 단추의 Click 이벤트 프로시저 작성하기 



052

[수정 사항 등록]을 클릭하면 ‘고객상세정보’ 프레임 안에 있는 각 컨트롤의 값이 [고객DB] 시

트에 업데이트되도록 하는 코드를 입력합니다.
36  

1    Private Sub cmd수정_Click()

2    Call 고객정보검색

3       셀위치.Offset(0, 4) = Me.txt조회성명

4       셀위치.Offset(0, 1) = Me.txt조회유형

5       셀위치.Offset(0, 2) = Me.txt조회계약일자

6       셀위치.Offset(0, 3) = Me.txt조회공사일자

7       셀위치.Offset(0, 5) = Me.txt조회번호

8       셀위치.Offset(0, 6) = Me.txt조회주소

9       셀위치.Offset(0, 7) = Me.txt조회메일

10      셀위치.Offset(0, 8) = Me.txt조회특기

11       셀위치.Offset(0, 17) = Val(Me.txt조회공사비)

12       셀위치.Offset(0, 18) = Me.txt조회결제방법    

13    Call cmd검색_Click

14    MsgBox Me.txt조회성명 & "님의 정보가 수정되었습니다"

End Sub

1   ‘cmd수정’ 명령 단추를 클릭했을 때 발생하는 이벤트 프로시저를 시작합니다.

2   ‘고객정보검색’ 프로시저를 호출합니다. 이 프로시저는 [검색프로그램] 모듈에 작성되어 있으며 이 프로

시저가 실행되고 나면 ‘셀위치’ 변수에는 목록 상자에서 선택한 고객의 고객코드가 [고객DB] 시트의 어

느 셀에 있는지 셀 정보를 가지고 복귀합니다.

3   ~ 12  각 텍스트 상자의 현재값을 [고객DB] 시트의 해당 셀에 다시 입력합니다. 

입력

13   변경된 데이터로 다시 검색하기 위하여 ‘cmd검색_Click’ 이벤트 프로시저를 실행합니다.

14   정보가 수정되었다는 메시지 대화상자를 나타냅니다.   

[개체]에서 [cmd삭제]를 클릭합니다. 

‘cmd삭제_Click’ 이벤트 프로시저가 추가됩니다.
37 고객 삭제 단추의 Click 이벤트 프로시저 작성하기 

[고객 삭제]를 클릭하면 선택된 고객 정보가 [고객DB] 시트에서 삭제되도록 하는 코드를 입력

합니다.
38 

1    Private Sub cmd삭제_Click()

2        If Me.txt조회성명 = "" Then

3            MsgBox "삭제할 고객이 선택되지 않았습니다"

입력

4            Exit Sub

5        End If    

6        If MsgBox(Me.txt조회성명 & "님의 정보를 삭제하시겠습니까?", vbYesNo) = vbNo Then 

               Exit Sub

7    Call 고객정보검색

8        셀위치.EntireRow.Delete (xlUp)

9            Me.txt조회성명 = ""

10           Me.txt조회유형 = ""

11            Me.txt조회계약일자 = ""

12            Me.txt조회공사일자 = ""

13            Me.txt조회번호 = ""

14            Me.txt조회공사비 = ""

15            Me.txt조회결제방법 = ""

16            Me.txt조회주소 = ""

17            Me.txt조회메일 = ""

18            Me.txt조회특기 = ""

19            Me.lst조회항목.Clear

20           Me.img조회사진.Picture = Nothing    

21        Call cmd검색_Click

End Sub

1   ‘cmd삭제’ 명령 단추를 클릭했을 때 발생하는 이벤트 프로시저를 시작합니다.

2   삭제할 고객이 선택되었는지 확인하는 조건문입니다. 만약 목록 상자에서 삭제할 고객이 선택되지 않

은 상태라면 ‘고객상세정보’ 프레임에 있는 ‘txt조회성명’ 텍스트 상자가 비어 있습니다. 조건식이 참이

면 3  ~ 4를 실행합니다.

3   ~ 4  조건식이 참일 때 실행할 문장으로, 삭제할 고객이 선택되지 않았다는 메시지 대화상자를 나타내

고 프로시저를 종료합니다.

5   If 조건문을 종료합니다.

6   한 줄로 입력합니다. 삭제 진행여부를 확인하는 메시지 대화상자를 나타내고, [아니오]를 클릭하면 프

로시저가 종료됩니다. 한 줄에 입력한 If 조건문으로, End If 문이 없습니다. 

7   ‘고객정보검색’ 프로시저를 호출합니다. 이 프로시저는 [검색프로그램] 모듈에 작성되어 있으며 이 프로

시저가 실행되고 나면 ‘셀위치’ 변수에는 목록 상자에서 선택한 고객의 고객코드가 [고객DB] 시트의 어

느 셀에 있는지 셀 정보를 가지고 복귀합니다.

8   ‘셀위치’ 변수에 저장된 셀의 행 전체를 삭제합니다.

9   ~ 20 ‘상세고객정보’ 프레임의 컨트롤들을 모두 초기화합니다.

21   삭제된 결과를 확인하기 위하여 ‘cmd검색_Click’ 이벤트 프로시저를 실행합니다.

같은 방법으로 [개체]에

서 [cmd고객DB], [cmd닫기], [cmd등록]을 각각 클릭하여 ‘Click’ 이벤트 프로시저를 작성합니다.
39 신규 등록, 고객 DB, 닫기 단추의 Click 이벤트 프로시저 작성하기 

1    Private Sub cmd고객DB_Click()

2       Sheets("고객DB").Select

End Sub

1    Private Sub cmd닫기_Click()

2       Unload Me

End Sub

1    Private Sub cmd등록_Click()

2        Unload Me

3       frm등록.Show vbModeless

End Sub

1   ‘cmd고객DB’ 명령 단추를 클릭했을 때 실행할 이벤트 프로시저를 시작합니다.

2   [고객DB] 시트를 선택합니다.

1   ‘cmd닫기’ 명령 단추를 클릭했을 때 실행할 이벤트 프로시저를 시작합니다.

2   현재 폼을 닫습니다.

입력

1   ‘cmd등록’ 명령 단추를 클릭했을 때 실행할 이벤트 프로시저를 시작합니다.

2   현재 폼을 닫습니다.

3   [고객 등록] 폼을 모덜리스로 열기합니다.

1  [Visual Basic 편집기]의 [표준] 도구 모음에서 [Sub/사용자 정의 폼 실행

]을 클릭합니다. 폼이 실행되면 2  [검색하기]를 클릭합니다. 3  목록 상자에 고객 정보가 표시되

면 한 고객을 클릭합니다. ‘고객상세정보’ 프레임에 각 항목이 표시됩니다.

40 결과 확인하기 

1

2

3



057

1  ‘고객상세정보’ 프레임의 내용을 수정한 후 [수정 사항 등록]을 클릭합니다. 수정되었다는 메

시지가 나타나면 2  [확인]을 클릭합니다. 3  목록 상자에서 삭제할 고객을 클릭한 후 4  [고객 삭

제]를 클릭합니다. 삭제할 것인지 확인하는 메시지가 나타나면 5  [예]를 클릭합니다. 

41 

1  [삽입]-[모듈]을 클릭합니다. 2  [속성] 창에서 [이름]을 메뉴단

추로 변경합니다.
42 폼 실행 프로시저 작성하기 

실력향상 [메뉴단추] 모듈에는 [고객등록] 폼과 [고객 검색/삭제] 폼을 실행하는 프로시저와 [고객DB] 시트를 선택하고 프로그램

을 종료하는 프로시저를 작성합니다. 작성된 프로시저는 [메뉴] 시트에서 단추(양식 컨트롤)에 연결합니다.

2

1

3

5

4

1

2 입력

[메뉴단추] 모듈의 [코드] 창에 다음과 같이 네 개의 프로시저를 작성합니다.43 

Sub 고객등록()

1      frm등록.Show vbModeless

End Sub

Sub 고객검색()

1      frm검색.Show vbModeless

End Sub

Sub 고객DB()

1      Sheets("고객DB").Select

End Sub

Sub 프로그램종료()

1      If MsgBox("프로그램을 저장한 후 종료하시겠습니까?", vbYesNo) = vbYes Then

2            ThisWorkbook.Save

3            Application.Quit

4      End If

End Sub

1   [고객등록] 폼을 모덜리스로 실행합니다.

1   [고객 검색/삭제] 폼을 모덜리스로 실행합니다.

1   [고객DB] 시트를 선택합니다.

1   메시지 대화상자를 나타내고, [예]를 클릭하면 2  ~ 3이 실행되도록 하는 조건문입니다.

2   현재 통합 문서를 저장합니다.

입력

3   엑셀 프로그램을 종료합니다.

4   조건문을 종료합니다.

1  [메뉴] 시트 탭을 클릭합니다. 2  [개발 도구] 탭-[삽입] 그룹-[단

추(양식 컨트롤) ]을 클릭한 후 3  제목 아래 빈 공간에 드래그하여 삽입합니다. [매크로 지정] 대

화상자의 4  [매크로 이름]에서 [고객등록]을 클릭한 후 5  [확인]을 클릭합니다.

44 단추에 프로시저 연결하기 

3 드래그

2

1

4

5

1  단추의 레이블을 신규 고객 등록으로 변경합니다. [신규 고객 등록] 단추를 Ctrl + Shift 를 

누른 채 드래그하여 세 개를 복사합니다.
45  

1 입력

2 Ctrl + Shift +드래그 3회

1  두 번째 단추를 마우스 오른쪽 버튼으로 클릭한 후 2  [매크로 지정]을 클릭합니다. 3  [매

크로 지정] 대화상자의 [매크로 이름]에서 4  [고객검색]을 클릭한 후 5  [확인]을 클릭합니다. 
46  

2

1 마우스 오른쪽 버튼 클릭

3

4

1  단추의 레이블을 고객 검색 및 수정으로 변경합니다. 2  같은 방법으로 나머지 단추에도 매

크로를 연결하고 레이블을 변경합니다.
47  

1 입력

2




